

Paterson, R. A. (2012). Constructing applicative functors. Paper presented at the 11th International Conference, Mathematics of Program Construction, 25 - 27 Jun 2012, Madrid, Spain.



**CITY UNIVERSITY  
LONDON**

[City Research Online](#)

**Original citation:** Paterson, R. A. (2012). Constructing applicative functors. Paper presented at the 11th International Conference, Mathematics of Program Construction, 25 - 27 Jun 2012, Madrid, Spain.

**Permanent City Research Online URL:** <http://openaccess.city.ac.uk/1141/>

#### **Copyright & reuse**

City University London has developed City Research Online so that its users may access the research outputs of City University London's staff. Copyright © and Moral Rights for this paper are retained by the individual author(s) and/ or other copyright holders. All material in City Research Online is checked for eligibility for copyright before being made available in the live archive. URLs from City Research Online may be freely distributed and linked to from other web pages.

#### **Versions of research**

The version in City Research Online may differ from the final published version. Users are advised to check the Permanent City Research Online URL above for the status of the paper.

#### **Enquiries**

If you have any enquiries about any aspect of City Research Online, or if you wish to make contact with the author(s) of this paper, please email the team at [publications@city.ac.uk](mailto:publications@city.ac.uk).

# Constructing Applicative Functors

Ross Paterson

City University London, UK

**Abstract.** Applicative functors define an interface to computation that is more general, and correspondingly weaker, than that of monads. First used in parser libraries, they are now seeing a wide range of applications. This paper sets out to explore the space of non-monadic applicative functors useful in programming. We work with a generalization, lax monoidal functors, and consider several methods of constructing useful functors of this type, just as transformers are used to construct computational monads. For example, coends, familiar to functional programmers as existential types, yield a range of useful applicative functors, including left Kan extensions. Other constructions are final fixed points, a limited sum construction, and a generalization of the semi-direct product of monoids. Implementations in Haskell are included where possible.

## 1 Introduction

This paper is part of a tradition of applying elementary category theory to the design of program libraries. Moggi [16] showed that the notion of monad could be used to structure denotational descriptions of programming languages, an idea carried over to program libraries by Wadler [20]. It turns out that the monads useful in semantics and programming can be constructed from a small number of monad transformers also identified by Moggi [17].

Applicative functors [15] provide a more limited interface than monads, but in return have more instances. All monads give rise to applicative functors, but our aim is to explore the space of additional instances with applications to programming. We are particularly interested in general constructions, with which programmers can build their own applicative functors, knowing that they satisfy the required laws. It is already known that applicative functors, unlike monads, can be freely composed. We identify a number of further general constructions, namely final fixed points, a limited sum construction, a generalization of semi-direct products of monoids, and coends (including left Kan extensions). By combining these constructions, one can obtain most of the computational applicative functors in the literature, with proofs of their laws. General constructions also clarify the relationships between seemingly unrelated examples, and suggest further applications.

Elementary category theory provides an appropriately abstract setting for the level of generality we seek. An idealized functional language corresponds to a type of category with first-class functions (a cartesian closed category). Applicative functors on such a category are equivalent to a simpler form called lax monoidal

functors, which are more convenient to work with. We can build up lax monoidal functors in more general ways by ranging across several different categories, as long as the end result acts on the category of our functional language, and is thus applicative. Familiarity with the basic definitions of categories and functors is assumed. The other notions used are mostly shallow, and will be explained along the way.

In the next section, we introduce applicative and lax monoidal functors. The rest of the paper describes the general constructions, illustrated with examples in Haskell where possible. Two proof styles are used throughout the paper. When making statements that apply to any category, we use standard commuting diagrams. However many statements assume a cartesian closed category, or at least a category with products. For these we use the internal language of the category, which provides a term language with equational reasoning that will be familiar to functional programmers.

## 2 Applicative Functors

The categorical notion of “functor” is modelled in Haskell with the type class

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Instances include a variety of computational concepts, including containers, in which `fmap` modifies elements while preserving shape. Another class of instances are “notions of computation”, including both monads and applicative functors, in which terms of type  $F\ a$  correspond to computations producing values of type  $a$ , but also having an “effect” described by the functor  $F$ , e.g. modifying a state, possibly throwing an exception, or non-determinism. The requirement that  $F$  be a functor allows one to modify the value returned without changing the effect.

The applicative interface adds pure computations (having no effect) and an operation to sequence computations, combining their results. It is described by a type class:

```
class Functor f => Applicative f where
  pure :: a -> f a
  (<*>) :: f (a -> b) -> f a -> f b
```

If we compare this with the type class of monads:

```
class Monad m where
  return :: a -> m a
  (>>=) :: m a -> (a -> m b) -> m b
```

we see that `pure` corresponds to `return`; the difference lies in the sequencing operations. The more powerful `>>=` operation available with monads allows the choice of the second computation to depend on the result of the first, while in the applicative case there can be no such dependency. Every monad can be made an applicative functor in a uniform way, here illustrated with the `Maybe` monad:

```
instance Functor Maybe where
  fmap f m = m >=> \ x -> return (f x)

instance Applicative Maybe where
  pure = return
  mf <*> mx = mf >=> \ f -> mx >=> \ x -> return (f x)
```

For functors that are also monads the monadic interface is often more convenient, but here we shall be more interested in applicative functors that are not also monads. A simple example is a constant functor returning a monoid [15]. Here is that functor expressed in Haskell using the `Monoid` class, which defines an associative binary operation `<>` with identity `mempty`:

```
newtype Constant a b = Constant a

instance Functor (Constant a) where
  fmap f (Constant x) = Constant x

instance Monoid a => Applicative (Constant a) where
  pure _ = Constant mempty
  Constant x <*> Constant y = Constant (x <> y)
```

The more limited applicative interface has many more instances, some of which will be presented in later sections. For example, the constrained form of sequencing offered by the applicative interface makes possible instances in which part of the value is independent of the results of computations, e.g. parsers that pre-generate parse tables [18]. Unlike monads, applicative functors are closed under composition.

However many applications of monads, such as traversal of containers, can be generalized to the applicative interface [15].

## 2.1 Lax Monoidal Functors

The applicative interface is convenient for programming, but in order to explore relationships between functors we shall use an alternative form with a more symmetrical sequencing operation:

```
class Functor f => Monoidal f where
  unit :: f ()
  mult :: f a -> f b -> f (a, b)
```

This interface, with identity and associativity laws, is equivalent to the applicative interface—the operations are interdefinable:

```
pure x = fmap (const x) unit
a <*> b = fmap (uncurry id) (mult a b)

unit = pure ()
mult a b = fmap (,) a <*> b
```

If we uncurry the operation `mult` of the `Monoidal` class, we obtain an operation  $\otimes : Fa \times Fb \rightarrow F(a \times b)$ . This suggests generalizing from products to other binary type constructors, a notion known in category theory as a monoidal category.

A *monoidal category* [13] consists of a category  $\mathcal{C}$ , a functor  $\otimes : \mathcal{C} \times \mathcal{C} \rightarrow \mathcal{C}$  and an object  $\top$  of  $\mathcal{C}$ , with coherent natural isomorphisms

$$\begin{aligned} \lambda : \top \otimes a &\cong a && \text{(left identity)} \\ \rho : a \otimes \top &\cong a && \text{(right identity)} \\ \alpha : a \otimes (b \otimes c) &\cong (a \otimes b) \otimes c && \text{(associativity)} \end{aligned}$$

A *symmetric monoidal category* also has

$$\sigma : a \otimes b \cong b \otimes a \quad \text{(symmetry)}$$

Both products and coproducts are examples of monoidal structures, and both are also symmetric. Given a monoidal category  $\langle \mathcal{C}, \top, \otimes, \lambda, \rho, \alpha \rangle$ , the category  $\mathcal{C}^{\text{op}}$ , obtained by reversing all the morphisms of  $\mathcal{C}$ , also has a monoidal structure:  $\langle \mathcal{C}^{\text{op}}, \top, \otimes, \lambda^{-1}, \rho^{-1}, \alpha^{-1} \rangle$ . The product of two monoidal categories is also monoidal, combining the isomorphisms of the two categories in parallel.

Often we simply refer to the category when the monoidal structure is clear from the context.

Some functors preserve this structure exactly, with  $\top' = F\top$  and  $Fa \otimes' Fb = F(a \otimes b)$ ; a trivial example is the identity functor. Others, such as the product functor  $\times : \mathcal{A} \times \mathcal{A} \rightarrow \mathcal{A}$  preserve it up to isomorphism:

$$\begin{aligned} 1 &\cong 1 \times 1 \\ (a_1 \times a_2) \times (b_1 \times b_2) &\cong (a_1 \times b_1) \times (a_2 \times b_2) \end{aligned}$$

We obtain a larger and more useful class of functors by relaxing further, requiring only morphisms between the objects in each pair, thus generalizing the class `Monoidal` above from products to any monoidal category.

A *lax monoidal functor* between monoidal categories  $\langle \mathcal{C}, \otimes, \top \rangle$  and  $\langle \mathcal{C}', \otimes', \top' \rangle$  consists of a functor  $F : \mathcal{C} \rightarrow \mathcal{C}'$  with natural transformations

$$\begin{aligned} u : \top' &\rightarrow F\top && \text{(unit)} \\ \otimes : Fa \otimes' Fb &\rightarrow F(a \otimes b) && \text{(multiplication)} \end{aligned}$$

such that the following diagrams commute:

$$\begin{array}{ccc} \top \otimes Fa & \xrightarrow{\lambda} & Fa \\ u \otimes Fa \downarrow & & \uparrow F\lambda \\ F\top \otimes Fa & \xrightarrow[\otimes]{} & F(\top \otimes a) \end{array} \qquad \begin{array}{ccc} Fa \otimes \top & \xrightarrow{\rho} & Fa \\ Fa \otimes u \downarrow & & \uparrow F\rho \\ Fa \otimes F\top & \xrightarrow[\otimes]{} & F(a \otimes \top) \end{array}$$

$$\begin{array}{ccccc} Fa \otimes (Fb \otimes Fc) & \xrightarrow{Fa \otimes \otimes} & Fa \otimes F(b \otimes c) & \xrightarrow[\otimes]{} & F(a \otimes (b \otimes c)) \\ \alpha \downarrow & & & & \downarrow F\alpha \\ (Fa \otimes Fb) \otimes Fc & \xrightarrow[\otimes \otimes Fc]{} & F(a \otimes b) \otimes Fc & \xrightarrow[\otimes]{} & F((a \otimes b) \otimes c) \end{array}$$

The first two diagrams state that  $u$  is the left and right identity respectively of the binary operation  $\otimes$ , while the last diagram expresses the associativity of  $\otimes$ .

## 2.2 Weak Commutativity

Although the definition of a lax monoidal functor neatly generalizes the **Monoidal** class, it lacks the counterpart of **pure**. We will also want an associated axiom stating that pure computations can be commuted with other computations. (There is a notion of symmetric lax monoidal functor, but requiring the ability to swap any two computations would exclude too many functors useful in computation, where the order in which effects occur is often significant.)

Thus we define an *applicative functor* on a symmetric monoidal category  $\mathcal{C}$  as consisting of a lax monoidal functor  $F : \mathcal{C} \rightarrow \mathcal{C}$ , with a natural transformation  $p : a \rightarrow F a$  (corresponding to the **pure** function of the **Applicative** class) satisfying  $p_{\top} = u$  and  $p \circ \otimes = \otimes \circ p \otimes p$ , plus a weak commutativity condition:

$$\begin{array}{ccccc}
 a \otimes F b & \xrightarrow{p \otimes F b} & F a \otimes F b & \xrightarrow{\otimes} & F(a \otimes b) \\
 \sigma \downarrow & & & & \downarrow F \sigma \\
 F b \otimes a & \xrightarrow{F b \otimes p} & F b \otimes F a & \xrightarrow{\otimes} & F(b \otimes a)
 \end{array}$$

We could also express the weak commutativity condition as a constraint on functors with a tensorial strength, but here we shall avoid such technicalities by assuming that function spaces are first-class types, with primitives to perform application and currying, or in categorical terms that we are working in a cartesian closed category (ccc). In particular, if  $\mathcal{A}$  is a ccc, any lax monoidal functor  $F : \mathcal{A} \rightarrow \mathcal{A}$  is also applicative. To show this, we make use of another advantage of working in a ccc, namely that we can conduct proofs in the internal  $\lambda$ -calculus of the category [12], in which variables of type  $a$  stand for arrows of  $\mathcal{A}(1, a)$ , and we write  $f(e_1, \dots, e_n)$  for  $f \circ \langle e_1, \dots, e_n \rangle$ . The result is a convenient language that is already familiar to functional programmers. When working in categories with products we shall calculate using the internal language; when products are not assumed we shall use diagrams.

In the internal language, we can define  $p : I \rightarrow F$  with the counterpart of the above definition of **pure** for any **Monoidal** functor:

$$p x = F(\text{const } x) u$$

The proof of weak commutativity is then a simple calculation in the internal language:

$$\begin{aligned}
F \sigma (p x \otimes y) &= F \sigma (F (const x) u \otimes y) && \text{definition of } p \\
&= F (\sigma \circ (const x) \times id) (u \otimes y) && \text{naturality of } \otimes \\
&= F (\sigma \circ (const x) \times id) (F \lambda^{-1} y) && \text{left identity} \\
&= F (id \times (const x) \circ \sigma) (F \lambda^{-1} y) && \text{naturality of } \sigma \\
&= F (id \times (const x) \circ \sigma \circ \lambda^{-1}) y && \text{functor} \\
&= F (id \times (const x) \circ \rho^{-1}) y && \text{symmetry} \\
&= F (id \times const x) (F \rho^{-1} y) && \text{functor} \\
&= F (id \times const x) (y \otimes u) && \text{right identity} \\
&= y \otimes F (const x) u && \text{naturality of } \otimes \\
&= y \otimes p x && \text{definition of } p
\end{aligned}$$

It is also known that lax monoidal functors in a ccc are equivalent to closed functors [5], which resemble the **Applicative** interface, but again the lax monoidal form is more convenient for defining derived functors.

Thus our strategy will be to construct a lax monoidal functor over the product structure of a ccc, but we may construct it from constituents involving other monoidal categories. As a simple example, we have seen that the product functor  $\times : \mathcal{A} \times \mathcal{A} \rightarrow \mathcal{A}$  is lax monoidal, and we can compose it with the diagonal functor from  $\mathcal{A}$  to  $\mathcal{A} \times \mathcal{A}$  (also lax monoidal) to obtain a lax monoidal functor from  $\mathcal{A}$  to  $\mathcal{A}$ :

$$F a = a \times a$$

though in this case the resulting functor is also monadic. In Section 5 we also use auxiliary categories with monoidal structures other than products.

### 3 Fixed Points, Limits and Colimits

A standard example of a computational monad is the list monad, which may be used to model backtracking. There is another lax monoidal functor on lists, with a unit constructing infinite lists and multiplication forming the zip of two lists [15]:

```

data ZipList a = Nil | Cons a (ZipList a)

instance Functor ZipList where
  fmap f Nil = Nil
  fmap f (Cons x xs) = Cons (f x) (fmap f xs)

instance Monoidal ZipList where
  unit = Cons () unit
  mult (Cons x xs) (Cons y ys) = Cons (x,y) (mult xs ys)
  mult _ _ = Nil

```

It turns out that this instance, and the proof that it satisfies the lax monoidal laws, follow from a general construction. We can observe that `ZipList` is a fixed point through the second argument of the binary functor  $F(a, b) = 1 + a \times b$ . That is,  $F$  is the functor  $\mathbf{Maybe} \circ \times$ , a composition of two lax monoidal functors and therefore lax monoidal.

There are two canonical notions of the fixed point of a functor, the initial and final fixed points, also known as data and codata. Initial fixed points can be used to define monads; here we use final fixed points to define lax monoidal functors. Recall that a parameterized final fixed point of a functor  $F : \mathcal{A} \times \mathcal{B} \rightarrow \mathcal{B}$  consists of a functor  $\nu F : \mathcal{A} \rightarrow \mathcal{B}$  with an isomorphism  $c : F(a, \nu F a) \cong \nu F a$  and an unfold operator  $[(\cdot)]$  constructing the unique morphism satisfying

$$\begin{array}{ccc} b & \xrightarrow{\quad [f] \quad} & \nu F a \\ f \downarrow & & \uparrow c \\ F(a, b) & \xrightarrow{\quad F(a, [f]) \quad} & F(a, \nu F a) \end{array}$$

If  $F$  is lax monoidal, we can define the unit and multiplication morphisms of a lax monoidal structure on  $\nu F$  as two of these unfolds:

$$\begin{array}{ccc} \top & \xrightarrow{\quad u_{\nu F} = [u_F] \quad} & \nu F \top \\ u_F \downarrow & & \uparrow c \\ F(\top, \top) & \xrightarrow{\quad F(\top, u_{\nu F}) \quad} & F(\top, \nu F) \end{array}$$

$$\begin{array}{ccc} \nu F a_1 \otimes \nu F a_2 & \xrightarrow{\quad \otimes_{\nu F} = [\otimes_F \circ c^{-1} \otimes c^{-1}] \quad} & \nu F (a_1 \otimes a_2) \\ c^{-1} \otimes c^{-1} \downarrow & & \uparrow c \\ F(a_1, \nu F a_1) \otimes F(a_2, \nu F a_2) & & \\ \otimes_F \downarrow & & \\ F(a_1 \otimes a_2, \nu F a_1 \otimes \nu F a_2) & \xrightarrow{\quad F(a_1 \otimes a_2, \otimes_{\nu F}) \quad} & F(a_1 \otimes a_2, \nu F (a_1 \otimes a_2)) \end{array}$$

In particular, for  $F = \mathbf{Maybe} \circ \times$ , this construction yields a lax monoidal functor equivalent to `ZipList` above.

One can prove using fusion that this definition does indeed satisfy the lax monoidal laws, but we shall prove a more general result instead.

### 3.1 Limits

Ignoring the parameter  $\mathcal{A}$  for the moment, another way to define the final fixed point of a functor  $F : \mathcal{B} \rightarrow \mathcal{B}$  starts with the terminal object  $1$ . Using with the



unique morphism  $!_{F1} : F1 \rightarrow 1$ , we can define a chain of objects and morphisms:

$$\cdots \longrightarrow F^3 1 \xrightarrow{F^2 !_{F1}} F^2 1 \xrightarrow{F !_{F1}} F 1 \xrightarrow{!_{F1}} 1$$

The final fixed point  $\nu F$  is defined as the limit of this chain, an object with a commuting family of morphisms (a *cone*) to the objects of the chain:

$$\begin{array}{c} \nu F \\ \swarrow \quad \searrow \quad \searrow \quad \searrow \\ \cdots \longrightarrow F^3 1 \xrightarrow{F^2 !_{F1}} F^2 1 \xrightarrow{F !_{F1}} F 1 \xrightarrow{!_{F1}} 1 \end{array}$$

such that any other such cone, say from an object  $B$ , can be expressed as a composition of a unique morphism  $B \rightarrow \nu F$  and the cone from  $\nu F$ .

This construction is sufficient for final fixed points of regular functors like **ZipList**, but for the general case we need to lift the whole thing to the category **Fun**( $\mathcal{A}, \mathcal{B}$ ), whose objects are functors  $\mathcal{A} \rightarrow \mathcal{B}$ , and whose morphisms are natural transformations. Given a functor  $\Phi$  on this category, we can repeat the above construction in the functor category, starting with the constant functor  $\bar{1}$ :

$$\begin{array}{c} \nu \Phi \\ \swarrow \quad \searrow \quad \searrow \quad \searrow \\ \cdots \longrightarrow \Phi^3 \bar{1} \xrightarrow{\Phi^2 !_{\Phi \bar{1}}} \Phi^2 \bar{1} \xrightarrow{\Phi !_{\Phi \bar{1}}} \Phi \bar{1} \xrightarrow{!_{\Phi \bar{1}}} \bar{1} \end{array}$$

A standard result holds that limits in **Fun**( $\mathcal{A}, \mathcal{B}$ ) may be constructed from point-wise limits in  $\mathcal{B}$  [13, p. 112].

On the way to defining the final fixed point of  $\Phi$  as a lax monoidal functor, we wish to require that  $\Phi$  preserve lax monoidal functors. To state this, we need a specialized notion of natural transformation for lax monoidal functors: a *monoidal transformation* between lax monoidal functors  $\langle F, \otimes, \top \rangle$  and  $\langle F', \otimes', \top' \rangle$  is a natural transformation  $h : F \rightarrow F'$  that preserves the lax monoidal operations:

$$\begin{array}{ccc} & \top & \\ u \swarrow & & \searrow u' \\ F \top & \xrightarrow{h} & F' \top \end{array} \qquad \begin{array}{ccc} F a \otimes F b & \xrightarrow{h \otimes h} & F' a \otimes F' b \\ \otimes \downarrow & & \downarrow \otimes' \\ F(a \otimes b) & \xrightarrow{h} & F'(a \otimes b) \end{array}$$

Then given monoidal categories  $\mathcal{A}$  and  $\mathcal{B}$ , we can define a category **Mon**( $\mathcal{A}, \mathcal{B}$ ) with lax monoidal functors as objects and monoidal transformations between them as morphisms. Now suppose we have a diagram in **Mon**( $\mathcal{A}, \mathcal{B}$ ), e.g. the chain

$$\cdots \longrightarrow F_3 \xrightarrow{f_2} F_2 \xrightarrow{f_1} F_1 \xrightarrow{f_0} F_0$$

We can construct a limit in  $\mathbf{Fun}(\mathcal{A}, \mathcal{B})$  (from pointwise limits in  $\mathcal{B}$ ):

$$\begin{array}{c}
 F \\
 \swarrow t_3 \quad \searrow t_0 \\
 \cdots \longrightarrow F_3 \xrightarrow{f_2} F_2 \xrightarrow{f_1} F_1 \xrightarrow{f_0} F_0
 \end{array}$$

To extend  $F$  to a limit of this diagram in  $\mathbf{Mon}(\mathcal{A}, \mathcal{B})$ , we want to define operations  $u$  and  $\otimes$  on  $F$  such that the  $t_i$  are monoidal transformations, i.e. satisfying the following equations in  $\mathcal{B}$ :

$$\begin{array}{ccc}
 & \top & \\
 u \swarrow & & \searrow u_i \\
 F \top & \xrightarrow{t_i} & F_i \top
 \end{array}
 \qquad
 \begin{array}{ccc}
 F a \otimes F b & \xrightarrow{t_i \otimes t_i} & F_i a \otimes F_i b \\
 \otimes \downarrow & & \downarrow \otimes_i \\
 F(a \otimes b) & \xrightarrow{t_i} & F_i(a \otimes b)
 \end{array}$$

These equations imply that  $u$  and  $\otimes$  are mediating morphisms to the limits in  $\mathcal{B}$ , and thus uniquely define them. It remains to show that  $\otimes$  is a natural transformation, and that  $u$  and  $\otimes$  satisfy the identity and associativity laws. Each of these four equations is proven in the same way: we show that the two sides of the equation are equalized by each  $t_i$ , as a consequence of the corresponding equation on  $F_i$ , and thus, by universality, must be equal. For example, for the left identity law we have the diagram

$$\begin{array}{ccccc}
 \top \otimes F a & \xrightarrow{\lambda} & & & F a \\
 \downarrow u \otimes F a & \searrow \top \otimes t_i & \top \otimes F_i a & \xrightarrow{\lambda} & F_i a \\
 & & \downarrow u_i \otimes F_i a & & \uparrow F_i \lambda \\
 & & F_i \top \otimes F_i a & \xrightarrow{\otimes_i} & F_i(\top \otimes a) \\
 & \nearrow t_i \otimes t_i & & \nwarrow t_i & \\
 F \top \otimes F a & \xrightarrow{\otimes} & & & F(\top \otimes a)
 \end{array}$$

The central panel is the left identity law for  $F_i$ , while the four surrounding panels follow from the definitions of  $u$  and  $\otimes$  and the naturality of  $\lambda$  and  $t_i$ . Thus the two morphisms  $\top \otimes F a \rightarrow F a$  on the perimeter of the diagram is equalized by  $t_i$ . Since the universality of the limit implies that such a morphism is unique, they must be equal. We have proven:

**Proposition 1.** *If  $\mathcal{B}$  is complete, then so is  $\mathbf{Mon}(\mathcal{A}, \mathcal{B})$ .*

Applying this to the chain of the fixed point construction, we have the immediate corollary that the final fixed point of a higher-order functor  $\Phi$  on lax

monoidal functors is a uniquely determined extension of the final fixed point of  $\Phi$  on ordinary functors. For example, `ZipList` is the final fixed point of the higher-order functor  $\Phi$  is defined by  $\Phi Z a = \text{Maybe}(a \times Z a)$ .

### 3.2 Sums

The dual notion, colimits, is not as easily handled. We can construct sums in special cases, such as adding the identity functor to another lax monoidal functor:

```
data Lift f a = Return a | Others (f a)

instance Functor f => Functor (Lift f) where
  fmap f (Return x) = Return (f x)
  fmap f (Others m) = Others (fmap f m)

instance Monoidal f => Monoidal (Lift f) where
  unit = Return ()
  mult (Return x) (Return y) = Return (x, y)
  mult (Return x) (Others my) = Others (fmap ((,) x) my)
  mult (Others mx) (Return y) = Others (fmap (flip (,) y) mx)
  mult (Others mx) (Others my) = Others (mult mx my)
```

Here pure computations (represented by the identity functor and the constructor `Return`) may be combined with `mult`, but are converted to the other functor if either computation involves that functor.

Applying this construction to the constant functor yields a form of computations with exceptions that collects errors instead of failing at the first error [4, 15]:

```
type Except err a = Lift (Constant err) a
```

That is, in a computation `mult e1 e2`, after a failure in `e1`, the whole computation will fail, but not before executing `e2` in case it produces errors that should be reported together with those produced by `e1`.

The fixed point  $L \cong \text{Lift}(I \times L)$  expands to non-empty lists combined with a “long zip”, in which the shorter list is padded with copies of its last element to pair with the remaining elements of the longer list, as suggested by Jeremy Gibbons and Richard Bird<sup>1</sup>:

```
data PadList a = Final a | NonFinal a (PadList a)

instance Functor PadList where
  fmap f (Final x) = Final (f x)
  fmap f (NonFinal x xs) = NonFinal (f x) (fmap f xs)
```

---

<sup>1</sup> Personal communication, 5 July 2011.

```

instance Monoidal PadList where
  unit = Final ()
  mult (Final x) (Final y) = Final (x, y)
  mult (Final x) (NonFinal y ys) =
    NonFinal (x, y) (fmap ((,) x) ys)
  mult (NonFinal x xs) (Final y) =
    NonFinal (x, y) (fmap (flip (,) y) xs)
  mult (NonFinal x xs) (NonFinal y ys) =
    NonFinal (x, y) (mult xs ys)

```

A straightforward generalization is  $L \cong \mathbf{Lift}(I \times (F \circ L))$  for any lax monoidal  $F$ , defining forms of long zip for various kinds of tree.

The **Lift** construction is able to combine computations in the identity functor with those of another lax monoidal functor  $F$  because there is a monoidal transformation between the two, namely the arrow **pure**. We can generalize:

**Proposition 2.** *If  $\mathcal{J}$  is an upper semi-lattice and  $\mathcal{B}$  is a ccc with finite coproducts, a diagram  $\Delta : \mathcal{J} \rightarrow \mathbf{Mon}(\mathcal{A}, \mathcal{B})$  has a colimit.*

*Proof.* Define a functor  $F$  by

$$F a = \sum_{j \in \mathcal{J}} C_j (\Delta_j a)$$

$$F f (C_j x) = C_j (f x)$$

where the  $C_j : \Delta_j \rightarrow F$  are tagging injections (constructors) marking the terms of the sum. Then we can define a lax monoidal structure on  $F$  as follows:

$$u = C_{\perp} u_{\perp}$$

$$C_j a \otimes C_k b = C_{j \sqcup k} (\Delta_{j \leq j \sqcup k} a, \Delta_{k \leq j \sqcup k} b)$$

Naturality of  $\otimes$  and the identity and associativity laws follow from simple calculations.  $\square$

For example, in the case of **Lift**,  $\mathcal{J}$  is a two-element lattice  $0 \leq 1$ , with  $\Delta_0 = I$ ,  $\Delta_1 = F$  and  $\Delta_{0 \leq 1} = p$ .

## 4 Generalized Semi-direct Products

A pioneering instance of the applicative interface was the parser combinator library of Swierstra and Duponcheel [18], which we here rehearse in a greatly cut-down form.

These parsers are applied to the output of lexical analysis. Given a type **Symbol** enumerating symbol types, parser input consists of a list of **Tokens**, recording the symbol type and its text:

```

type Token = (Symbol, String)

```

For example, there might be a `Symbol` for numeric literals, in which case the corresponding `String` would record the text of the number. Parsers take a list of tokens and return either an error string or a parsed value together with the unparsed remainder of the input:

```
newtype Parser a = P ([Token] -> Either String (a, [Token]))
```

This type is a monad (built by adding to an exception monad a state consisting of a list of tokens), and therefore also an applicative functor. Parsers can be built using primitives to peek at the next symbol, to move to the next token returning the string value of the token read, and to abort parsing reporting an error:

```
nextSymbol :: Parser Symbol
advance    :: Parser String
throwError :: String -> Parser a
```

In order to construct recursive descent parsers corresponding to phrases of a grammar, one needs to keep track of whether a phrase can generate the empty string, and also the set of symbols that can begin a phrase (its *first* set). Swierstra and Duponcheel's idea was to define a type containing this information about a phrase, from which a deterministic parser for the phrase could be constructed:

```
data Phrase a = Phrase (Maybe a) (Map Symbol (Parser a))
```

The two components are:

- The type `Maybe a` indicates whether the phrase can generate the empty string, and if so provides a default output value.
- The type `Map Symbol (Parser a)` records which symbols can start the phrase, and provides for each a corresponding deterministic parser.

The `Functor` instance for this type follows from the structure of the type:

```
instance Functor Phrase where
  fmap f (Phrase e t) = Phrase (fmap f e) (fmap (fmap f) t)
```

The idea, then, is to build a value of this type for each phrase of the grammar, with the following conversion to a deterministic parser:

```
parser :: Phrase a -> Parser a
parser (Phrase e t)
  | null t = def
  | otherwise = do
    s <- nextSymbol
    findWithDefault def s t
  where
    def = case e of
      Just x -> return x
      Nothing -> throwError ("expected " ++ show (keys t))
```

A parser for a single symbol, returning its corresponding text, is

```
symbol :: Symbol -> Phrase String
symbol s = Phrase Nothing (singleton s advance)
```

Alternatives are easily built:

```
(<|>) :: Phrase a -> Phrase a -> Phrase a
Phrase e1 t1 <|> Phrase e2 t2 =
  Phrase (e1 'mplus' e2) (t1 'union' t2)
```

In a realistic library, one would want to check that at most one of the alternatives could generate the empty string, and that the first sets were disjoint. The information in the **Phrase** type makes it possible to determine this check before parsing, but we omit this in our simplified presentation.

Now the lax monoidal structure corresponds to the empty phrase and concatenation of phrases. A phrase  $\alpha\beta$  can generate the empty string only if both the constituent phrases can, but the emptiness information for  $\alpha$  also determines whether the initial symbols of  $\alpha\beta$  include those of  $\beta$  in addition to those of  $\alpha$ :

```
instance Monoidal Phrase where
  unit = Phrase unit empty
  mult (Phrase e1 t1) (~p2@(Phrase e2 t2)) =
    Phrase (mult e1 e2) (union t1' t2')
  where
    t1' = fmap ('mult' parser p2) t1
    t2' = maybe empty (\ x -> fmap (fmap ((,) x)) t2) e1
```

In Haskell, a tilde marks a pattern as lazy, meaning it is not matched until its components are used. It is used here so that **Phrase** values can be recursively defined, as long as one avoids left recursion.

We might wonder whether this definition is an instance of a general construction. We note that the **Phrase** type is a pair, and the first components are combined using the lax monoidal operations on **Maybe**, independent of the second components. This is similar to a standard construction on monoids, the semi-direct product, which takes a pair of monoids  $\langle A, *, 1 \rangle$  and  $\langle X, +, 0 \rangle$  with an action  $(\cdot) : A \times X \rightarrow X$ , and defines a monoid on  $A \times X$ , with binary operation

$$(a, x) \odot (b, y) = (a * b, x + (a \cdot y))$$

and identity  $(1, 0)$ . For example Horner's Rule for the evaluation of a polynomial  $a_n x^n + \dots + a_1 x + a_0$  can be expressed as a fold of such an operation over the list  $[(x, a_0), (x, a_1), \dots, (x, a_n)]$ , with the immediate consequence that the calculation can be performed in parallel (albeit with repeated calculation of the powers of  $x$ ).

We shall consider a generalization of the semi-direct product on lax monoidal functors, requiring

- a lax monoidal functor  $\langle F, \otimes, u \rangle : \langle \mathcal{A}, \otimes, \top \rangle \rightarrow \langle \mathcal{B}, \times, 1 \rangle$

- a functor  $G : \mathcal{A} \rightarrow \mathcal{B}$  with a natural family of monoids  $\oplus : G a \times G a \rightarrow G a$  and  $\emptyset : 1 \rightarrow G a$ .
- an operation  $\rtimes : G a \times (F b \times G b) \rightarrow G(a \otimes b)$  distributing over  $G$ :

$$\emptyset \rtimes q = \emptyset \quad (1)$$

$$(x \oplus y) \rtimes q = (x \rtimes q) \oplus (y \rtimes q) \quad (2)$$

- an operation  $\ltimes : (F a \times G a) \times G b \rightarrow G(a \otimes b)$  distributing over  $G$ :

$$p \ltimes \emptyset = \emptyset \quad (3)$$

$$p \ltimes (x \oplus y) = (p \ltimes x) \oplus (p \ltimes y) \quad (4)$$

also satisfying

$$(p \ltimes y) \rtimes r = p \ltimes (y \rtimes r) \quad (5)$$

**Proposition 3.** *Given the above functors and operations, there is a lax monoidal functor  $\langle H, \otimes_H, u_H \rangle : \langle \mathcal{A}, \otimes, \top \rangle \rightarrow \langle \mathcal{B}, \times, 1 \rangle$  defined by*

$$\begin{aligned} H a &= F a \times G a \\ u_H &= (u, \emptyset) \\ (a, x) \otimes_H (b, y) &= (a \otimes b, (x \rtimes (b, y)) \oplus ((a, x) \ltimes y)) \end{aligned}$$

provided that  $\ltimes$  and  $\rtimes$  are left and right actions on  $G$ , i.e.

$$u_H \ltimes z = z \quad (6)$$

$$(p \otimes_H q) \ltimes z = p \ltimes (q \rtimes z) \quad (7)$$

$$x \rtimes u_H = x \quad (8)$$

$$x \rtimes (q \otimes_H r) = (x \rtimes q) \rtimes r \quad (9)$$

*Proof.* It follows from their definitions that  $H$  is a functor and  $\otimes_H$  a natural transformation. Next, we show that  $u_H$  is the left and right identity of  $\otimes_H$ :

$$\begin{aligned} u_H \otimes_H (b, y) &= (1 \otimes b, (\emptyset \rtimes (b, y)) \oplus (u_H \ltimes y)) && \text{definition of } \otimes_H, u_H \\ &= (1 \otimes b, \emptyset \oplus y) && \text{equations (1) and (6)} \\ &= (b, y) && \text{monoid laws} \end{aligned}$$

$$\begin{aligned} (a, x) \otimes_H u_H &= (a \otimes 1, (x \rtimes u_H) \oplus ((a, x) \ltimes \emptyset)) && \text{definition of } \otimes_H, u_H \\ &= (a \otimes 1, x \oplus \emptyset) && \text{equations (8) and (3)} \\ &= (a, x) && \text{monoid laws} \end{aligned}$$

Finally, we must show that  $\otimes_H$  is associative:

$$\begin{aligned}
& ((a, x) \otimes_H (b, y)) \otimes_H (c, z) \\
&= (a \otimes b \otimes c, (((x \times (b, y)) \oplus ((a, x) \times y)) \times (c, z)) \oplus \\
&\quad (((a, x) \otimes_H (b, y)) \times z)) \quad \text{definition of } \otimes_H \\
&= (a \otimes b \otimes c, ((x \times (b, y)) \times (c, z)) \oplus (((a, x) \times y) \times (c, z)) \oplus \\
&\quad (((a, x) \otimes_H (b, y)) \times z)) \quad \text{equation (2)} \\
&= (a \otimes b \otimes c, (x \times ((b, y) \otimes_H (c, z))) \oplus \\
&\quad ((a, x) \times (y \times (c, z))) \oplus ((a, x) \times ((b, y) \times z))) \quad (9), (5) \text{ and } (7) \\
&= (a \otimes b \otimes c, (x \times ((b, y) \otimes_H (c, z))) \oplus \\
&\quad ((a, x) \times (y \times (c, z))) \oplus ((b, y) \times z))) \quad \text{equation (4)} \\
&= (a, x) \otimes_H ((b, y) \otimes_H (c, z)) \quad \text{definition of } \otimes_H
\end{aligned}$$

□

In particular, Proposition 3 identifies the properties we need to establish to demonstrate that **Phrase** is lax monoidal, and thus applicative.

## 5 Existentials and Coends

Many applicative functors are constructed using existential quantification, hiding a private representation type. We shall consider the corresponding categorical notion, called a coend.

Abbott, Altenkirch and Ghani [1] consider containers of the form

$$L\ c = \exists m. K\ m \rightarrow c$$

Here  $m$  is drawn from a set of shapes  $\mathcal{M}$  (a discrete category), the functor  $K : \mathcal{M} \rightarrow \mathcal{C}$  assigns to each shape a set of positions within containers of that shape, and the function provides a value for each of these positions. For example, the shape of a list is a natural number  $n$  giving its length, which  $K$  maps to the set of positions  $\{0, 1, \dots, n-1\}$ , the indices of the list.

There are several ways that we can extend container functors to obtain useful lax monoidal functors.

If we let  $\mathcal{M}$  be the natural numbers plus an upper bound  $\omega$ , and  $K\ n = \{i \mid i < n\}$ , then  $L$  represents finite and infinite lists. We can define lax monoidal operations:

$$\begin{aligned}
u &= (\omega, \text{const } \top) \\
(m, f) \otimes (n, g) &= (m \sqcap n, \langle f, g \rangle)
\end{aligned}$$

That is,  $u$  yields an infinite list, and  $\otimes$  constructs a list of pairs, whose length is the smaller of the lengths of the arguments. We recognize this functor as another version of the **ZipList** functor defined in Section 3. More generally, if  $\mathcal{M}$  has a monoidal structure that is a lower semi-lattice, and  $K\ (m_1 \sqcap m_2) \subseteq K\ m_i$ , then the lax monoidal structure on  $L$  computes zips on containers.



A type comprising arrays of different dimensions can be represented using a shape functor  $K$  satisfying  $K \top \cong 1$  and  $K(a \otimes b) \cong K a \times K b$ . Then we can define lax monoidal operations with  $u$  constructing a scalar and  $\otimes$  being cartesian product:

$$u = (\top, \text{const } ())$$

$$(m, f) \otimes (n, g) = (m \otimes n, f \times g)$$

We can approximate such multi-dimensional arrays using a Haskell existential type (specified by using the quantifier keyword `forall` before the data constructor):

```
data MultiArray a = forall i. Ix i => MA (Array i a)

instance Functor MultiArray where
  fmap f (MA a) =
    MA (array (bounds a) [(i, f e) | (i, e) <- assocs a])
```

The `unit` operation constructs a scalar, while `mult` forms the cartesian product of two arrays:

```
instance Monoidal MultiArray where
  unit = MA (array ((), ()) [((), ())])
  mult (MA xs) (MA ys) =
    MA (array ((lx, ly), (hx, hy))
      [(i, j), (x, y)) | (i, x) <- assocs xs,
                          (j, y) <- assocs ys])
  where
    (lx, hx) = bounds xs
    (ly, hy) = bounds ys
```

We could extend multi-dimensional arrays by adding a distinguished position, i.e. a cursor within the container:

$$L c = \exists m. K m \times (K m \rightarrow c)$$

When two arrays are combined with `mult`, their cursors are also paired to form a cursor on the product array.

Another example arises in Elliott's analysis of fusion [6], where folds are reified using a type

```
data FoldL b a = FoldL (a -> b -> a) a
```

The type constructor `FoldL` is not a functor, because its argument `a` occurs in both the domain and range of function types. Wishing to apply functorial machinery to these reified folds, Elliott introduced a related type that could be defined as a functor:

```
data WithCont z c = forall a. WC (z a) (a -> c)
```

```
instance Functor (WithCont z) where
  fmap g (WC z k) = WC z (g . k)
```

Although `FoldL` is not a functor, it nevertheless has operations similar to `unit` and `mult`. These can be described using a type class similar to `Monoidal`, but without the `Functor` superclass:

```
class Zip z where
  zunit :: z ()
  zmult :: z a -> z b -> z (a, b)
```

The above type constructor `FoldL` is an instance:

```
instance Zip (FoldL b) where
  zunit = FoldL const ()
  zmult (FoldL f1 z1) (FoldL f2 z2) =
    FoldL (\ (x,y) b -> (f1 x b, f2 y b)) (z1, z2)
```

This class is sufficient to define `WithCont` as a lax monoidal functor:

```
instance Zip z => Monoidal (WithCont z) where
  unit = WC zunit (const ())
  mult (WC t1 k1) (WC t2 k2) =
    WC (zmult t1 t2) (\ (x,y) -> (k1 x, k2 y))
```

## 5.1 Left Kan Extensions

We now consider the general case. Kan extensions are general constructions that have also found applications in programming. The right Kan has been used to construct generalized folds on nested types [10], to fuse types [7], and to construct a monad (the codensity monad) that can be used for program optimization [19, 8]. The lax monoidal functors discussed above are instances of the other variety, the left Kan.

Kan extensions have an elegant description at the level of functors. Given a functor  $K : \mathcal{M} \rightarrow \mathcal{C}$ , the left and right Kan extensions along  $K$  are defined as the left and right adjoints of the higher-order functor  $(\circ K)$  that maps to each functor  $\mathcal{C} \rightarrow \mathcal{A}$  to a functor  $\mathcal{M} \rightarrow \mathcal{A}$  [13]. That is, the left Kan extension a functor  $T : \mathcal{M} \rightarrow \mathcal{A}$  along  $K$  is a functor  $L : \mathcal{C} \rightarrow \mathcal{A}$  with a universal natural transformation  $\eta : T \rightarrow L \circ K$ :

$$\begin{array}{ccc}
 \mathcal{M} & \xrightarrow{K} & \mathcal{C} \\
 & \searrow T & \nearrow L \\
 & \mathcal{A} &
 \end{array}
 \quad \begin{array}{c}
 \eta \\
 \Rightarrow
 \end{array}$$

For our purposes, it will be more convenient to use the standard pointwise construction of the left Kan extension as a `coend`, corresponding to existential quantification in programming languages. For convenience, we assume that

the category  $\mathcal{A}$  is cartesian closed, and that  $\mathcal{C}$  is an  $\mathcal{A}$ -category [11], i.e. that the “hom-sets” of  $\mathcal{C}$  are objects of  $\mathcal{A}$ , with identity and composition morphisms satisfying the usual laws. Using the more familiar notation  $\exists$  in place of the integral sign favoured by category theorists, the left Kan extension of  $T : \mathcal{M} \rightarrow \mathcal{A}$  along  $K : \mathcal{M} \rightarrow \mathcal{C}$  is the functor  $L : \mathcal{C} \rightarrow \mathcal{A}$  defined by

$$L c = \exists m. T m \times_{\mathcal{C}} (K m, c)$$

The examples discussed above are instances of left Kan extensions:

- In the container example,  $T$  is the constant functor mapping to 1, and the monoidal structure on  $\mathcal{M}$  has  $\top = \omega$  and with  $\otimes$  as minimum.
- In the example of arrays with cursors,  $T$  is identified with  $K$ .
- In the `WithCont` example,  $\mathcal{M}$  is the subcategory of isomorphisms of  $\mathcal{A}$ .  $T$  can model any type constructor, as although type constructors (like `FoldL` above) need not be functorial, they still preserve isomorphisms.

To state that  $L c$  is a coend is to say that there is an initial dinatural transformation  $\omega : T m \times_{\mathcal{C}} (K m, c) \rightarrow L c$ . This dinaturality of  $\omega$  is expressed by the equation

$$\omega(T h x, k) = \omega(x, k \circ K h)$$

That is, the existentially qualified type  $m$  is abstract: we can change the representation without affecting the constructed value. The natural transformation  $\eta : T \rightarrow L \circ K$  is defined as

$$\eta x = \omega(x, id)$$

Initiality of  $\omega$  means that a natural transformation from  $L$  is uniquely determined by its action on terms of the internal language of the form  $\omega(x, k)$ . For example, we can define the action of  $L$  on arrows as

$$L f(\omega(x, k)) = \omega(x, f \circ k)$$

In order to make  $L$  lax monoidal, we shall assume that the functor  $K : \mathcal{M} \rightarrow \mathcal{C}$  is a *colax monoidal functor*, or equivalently a lax monoidal functor  $\mathcal{M}^{\text{op}} \rightarrow \mathcal{C}^{\text{op}}$ . That is, there are natural transformations

$$\begin{aligned} s &: K(a \otimes_{\mathcal{M}} b) \rightarrow K a \otimes_{\mathcal{C}} K b \\ n &: K \top_{\mathcal{M}} \rightarrow \top_{\mathcal{C}} \end{aligned}$$

such that the following diagrams commute:

$$\begin{array}{ccc} K(\top \otimes a) & \xrightarrow{s} & K \top \otimes K a \\ K \lambda \downarrow & & \downarrow n \otimes K a \\ K a & \xleftarrow{\lambda} & \top \otimes K a \end{array} \qquad \begin{array}{ccc} K(a \otimes \top) & \xrightarrow{s} & K a \otimes K \top \\ K \rho \downarrow & & \downarrow K a \otimes n \\ K a & \xleftarrow{\rho} & K a \otimes \top \end{array}$$

$$\begin{array}{ccccc}
K(a \otimes (b \otimes c)) & \xrightarrow{s} & K a \otimes K(b \otimes c) & \xrightarrow{K a \otimes s} & K a \otimes (K b \otimes K c) \\
\downarrow K \alpha & & & & \downarrow \alpha \\
K((a \otimes b) \otimes c) & \xrightarrow{s} & K(a \otimes b) \otimes K c & \xrightarrow{s \otimes K c} & (K a \otimes K b) \otimes K c
\end{array}$$

In the special case where the monoidal structure on  $\mathcal{C}$  is that of products, there is only one choice for  $n$ , namely the unique arrow  $K \top \rightarrow 1$ . Moreover in that case  $s : K(a \otimes b) \rightarrow K a \times K b$  can be broken down into two components:  $s = \langle s_1, s_2 \rangle$ .

**Proposition 4.** *If  $\mathcal{M}$  and  $\mathcal{C}$  are monoidal and  $\mathcal{A}$  has finite products,  $K$  is colax monoidal and  $T$  is lax monoidal, then  $L$  is lax monoidal, with*

$$\begin{aligned}
u_L &= \omega(u_T, n_K) \\
\omega(x_1, k_1) \otimes_L \omega(x_2, k_2) &= \omega(x_1 \otimes_T x_2, k_1 \times k_2 \circ s_K)
\end{aligned}$$

This is a special case of Proposition 5, which we shall prove in the next section.

A degenerate example has  $\mathcal{M}$  as the trivial category with one object and one morphism, so that  $T$  defines a monoid and  $K$  selects some object, with  $s_i = id$ . This describes computations that write output and also read an environment, but in which the output is independent of the environment:

$$L c = T \times (K \rightarrow c)$$

This applicative functor is a composition of two applicative functors that are also monads, but the composition is not a monad.

Another simple case arises when  $\mathcal{M}$  is a cartesian category, in which case an arbitrary functor  $K : \mathcal{M} \rightarrow \mathcal{C}$  can be made colax monoidal by setting  $s_i = K \pi_i$ . Thus we obtain the following Haskell version of the left Kan:<sup>2</sup>

```

data Lan t k c = forall m. Lan (t m) (k m -> c)

instance (Functor t, Functor k) => Functor (Lan t k) where
  fmap f (Lan x k) = Lan x (f . k)

instance (Monoidal t, Functor k) => Monoidal (Lan t k) where
  unit = Lan unit (const ())
  mult (Lan x1 k1) (Lan x2 k2) =
    Lan (mult x1 x2)
      (\ y -> (k1 (fmap fst y), k2 (fmap snd y)))

```

Although this implementation has the form of a general left Kan extension, it is limited to the Haskell category.

A richer example occurs in the modelling of behaviours of animations using applicative functors by Matlage and Gill [14]. The basic functor comprises a function over a closed interval of time, which can be modelled as pairs of times:

<sup>2</sup> In fact the `Functor` instance requires no assumptions about `t` and `k`, and in the `Monoidal` instance `Zip t` could replace `Monoidal t`.

```

data Interval = Between Time Time

instance Monoid Interval where
  mempty = Between inf (-inf)
  Between start1 stop1 <> Between start2 stop2 =
    Between (min start1 start2) (max stop1 stop2)

```

We would like to represent continuous behaviours by a type  $\exists i. K\ i \rightarrow T$ , for a functor  $K$  mapping pairs of times to closed intervals of time, with  $s_i$  mapping from larger intervals to smaller by truncation. We cannot express this directly in Haskell, which lacks dependent types, but we can approximate it with a type

```

data Behaviour a = B Interval (Time -> a)

```

provided we hide the representation and provide only an accessor function:

```

observe :: Behaviour a -> Time -> a
observe (B (Between start stop) f) t =
  f (max start (min stop t))

```

Now this type can be made monoidal with the following definitions, which preserve the abstraction:

```

instance Functor Behaviour where
  fmap f (B i g) = B i (f . g)

instance Monoidal Behaviour where
  unit = B mempty (const ())
  mult b1@(B i1 f1) b2@(B i2 f2) =
    B (i1 <> i2) (\ t -> (observe b1 t, observe b2 t))

```

The final functor used by Matlage and Gill can be obtained by adding constant behaviour using `Lift`:

```

type Active = Lift Behaviour

```

Thus a value of type `Active a` is either constant or a function of time over a given interval. A combination of such behaviours is constant only if both the arguments were.

## 5.2 The General Case

Our final example is a generalization of the type used by Baars, Löh and Swierstra [3] to construct parsers for permutations of phrases, which we express as

```

data Perms p a = Choice (Maybe a) [Branch p a]
data Branch p a = forall b. Branch (p b) (Perms p (b -> a))

```

This implementation is too subtle to explain in full detail here, but the `Perms` type is essentially an efficient representation of a collection of all the permutations of a set of elementary parsers (or actions, in other applications). The type in the original paper is equivalent to restricting our version of the `Perms` type to values of the forms `Choice (Just x) []` and `Choice Nothing bs`, allowing a single elementary parser to be added to the collection at a time. In contrast, the `mult` methods allows the interleaving of arbitrary collections of actions, allowing us to build them in any order.

The functor instances for these two types are straightforward:

```
instance Functor p => Functor (Perms p) where
  fmap f (Choice def bs) =
    Choice (fmap f def) (map (fmap f) bs)

instance Functor p => Functor (Branch p) where
  fmap f (Branch p perm) = Branch p (fmap (f .) perm)
```

Assuming that `p` is lax monoidal, we will construct instances for `Perms p` and `Branch p`. These types are mutually recursive, but we know that final fixed points preserve applicative functors.

We define an operator `***` as

```
(***) :: Monoidal f => f (a1 -> b1) -> f (a2 -> b2) ->
      f ((a1,a2) -> (b1,b2))
p *** q = fmap (\ (f,g) (x,y) -> (f x, g y)) (mult p q)
```

This is an example of the construction of static arrows from applicative functors [15]. Now, assuming that `Perms p` is lax monoidal, we can construct an instance for `Branch p` as a generalized left Kan extension:

```
instance Monoidal p => Monoidal (Branch p) where
  unit = Branch unit (pure id)
  mult (Branch p1 perm1) (Branch p2 perm2) =
    Branch (mult p1 p2) (perm1 *** perm2)
```

The instance for `Perms p` is constructed from the instance for `Branch p` as a generalized semi-direct product, which builds all the interleavings of the two collections of permutations:

```
instance Monoidal p => Monoidal (Perms p) where
  unit = Choice unit []
  mult (t1@(Choice d1 bs1)) (t2@(Choice d2 bs2)) =
    Choice (mult d1 d2)
      (map ('mult' include t2) bs1 ++
       map (include t1 'mult') bs2)
  where
    include :: Monoidal p => Perms p a -> Branch p a
    include p = Branch unit (fmap const p)
```

To encompass examples such as this, we need a generalization of the left Kan. Suppose the functor  $K$  factors through a monoidal category  $\mathcal{B}$ :

$$\begin{array}{ccc}
 & \mathcal{B} & \\
 H \nearrow & & \searrow J \\
 \mathcal{M} & \xrightarrow{K} & \mathcal{C} \\
 T \searrow & & \nearrow L \\
 & \mathcal{A} &
 \end{array}
 \quad \begin{array}{c}
 = \\
 \eta \\
 \text{dashed}
 \end{array}$$

We also assume a natural operator

$$\boxtimes : \mathcal{C}(J a, J b) \times \mathcal{C}(J c, J d) \rightarrow \mathcal{C}(J(a \otimes c), J(b \otimes d))$$

(corresponding to \*\*\* above) satisfying unit and associativity laws:

$$\begin{aligned}
 J \lambda \circ f \boxtimes \top &= f \circ J \lambda \\
 J \rho \circ \top \boxtimes f &= f \circ J \rho \\
 J \alpha \circ f \boxtimes (g \boxtimes h) &= (f \boxtimes g) \boxtimes h \circ J \alpha
 \end{aligned}$$

The situation of ordinary left Kan extensions is the special case where  $J$  is the identity functor and  $\boxtimes$  is  $\otimes_{\mathcal{C}}$ . However in general we do not require that  $\boxtimes$  be a functor. The key example of a structure with such an operator is an enriched premonoidal category, or “arrow” [2, 9].

**Proposition 5.** *If  $\mathcal{M}$  and  $\mathcal{B}$  are monoidal,  $\mathcal{A}$  has finite products,  $H$  is colax monoidal and  $T$  is lax monoidal, then  $F = L \circ J$  is lax monoidal, with*

$$\begin{aligned}
 F a &= \exists m. T m \times (J(H m) \rightarrow J a) \\
 F f(\omega(x, k)) &= \omega(x, J f \circ k) \\
 u_F &= \omega(u_T, J n_H) \\
 \omega(x_1, k_1) \otimes_F \omega(x_2, k_2) &= \omega(x_1 \otimes_T x_2, k_1 \boxtimes k_2 \circ J s_H)
 \end{aligned}$$

Instead of proving this directly, we show that the functor  $G : \mathcal{M}^{\text{op}} \times \mathcal{M} \times \mathcal{A}$  defined by

$$G(m', m, a) = T m \times (J(H m') \rightarrow J a)$$

is itself lax monoidal, and then use a general result about coends of lax monoidal functors. To see that  $G$  is lax monoidal, we note that  $T$  is lax monoidal, so we only need to show that the second component is. The left identity case is

$$\begin{aligned}
 F \lambda \circ id \boxtimes k \circ J(n_H \times 1 \circ s_H) &= k \circ J(\lambda \circ n_H \times 1 \circ s_H) && \text{left identity of } \boxtimes \\
 &= k \circ J(H \lambda) && \text{left identity of } H
 \end{aligned}$$

The right identity case is similar. Associativity relies on the associativity of  $\boxtimes$ :

$$\begin{aligned}
 J \alpha \circ k_1 \boxtimes (k_2 \boxtimes k_3 \circ J s_H) \circ J s_H & \\
 = J \alpha \circ k_1 \boxtimes (k_2 \boxtimes k_3) \circ J(id \times s_H \circ s_H) &&& \text{naturality of } \boxtimes \\
 = (k_1 \boxtimes k_2) \boxtimes k_3 \circ J(\alpha \circ id \times s_H \circ s_H) &&& \text{associativity of } \boxtimes \\
 = (k_1 \boxtimes k_2) \boxtimes k_3 \circ J(s_H \times id \circ s_H \circ H \alpha) &&& \text{associativity of } s_H
 \end{aligned}$$

Thus it suffices to show that coends preserve lax monoidal functors, which is our final result.

**Proposition 6.** *Given monoidal categories  $\mathcal{A}$  and  $\mathcal{B}$  and a ccc  $\mathcal{C}$ , with a lax monoidal functor  $G : \mathcal{A}^{\text{op}} \times \mathcal{A} \times \mathcal{B} \rightarrow \mathcal{C}$ , then the coend  $Fb = \exists a. G(a, a, b)$  is also lax monoidal, with*

$$\begin{aligned} Fb &= \exists a. G(a, a, b) \\ Ff(\omega x) &= \omega(G(id, id, f)x) \\ u_F &= \omega u_G \\ \omega x_1 \otimes_F \omega x_2 &= \omega(x_1 \otimes_G x_2) \end{aligned}$$

*Proof.* It is a standard result that a parameterized coend such as  $F$  defines a functor. Naturality of  $\otimes_F$  follows from naturality of  $\otimes_G$ :

$$\begin{aligned} F(f_1 \otimes f_2)(\omega x_1 \otimes_F \omega x_2) & \\ = F(f_1 \otimes f_2)(\omega(x_1 \otimes_G x_2)) & \text{definition of } \otimes_F \\ = \omega(G(id, id, f_1 \otimes f_2)(x_1 \otimes_G x_2)) & \text{definition of } F \\ = \omega(G(id, id, f_1)x_1 \otimes_G G(id, id, f_2)x_2) & \text{naturality of } \otimes_G \\ = \omega(G(id, id, f_1)x_1) \otimes_F \omega(G(id, id, f_2)x_2) & \text{definition of } \otimes_F \\ = Ff_1(\omega x_1) \otimes_F Ff_2(\omega x_2) & \text{definition of } F \end{aligned}$$

Similarly the left identity law for  $F$  follows from the corresponding law for  $G$ :

$$\begin{aligned} F\lambda(u_F \otimes_F \omega x) &= F\lambda(\omega u_G \otimes_F \omega x) & \text{definition of } u_F \\ &= F\lambda(\omega(u_G \otimes_G x)) & \text{definition of } \otimes_F \\ &= \omega(G(id, id, \lambda)(u_G \otimes_G x)) & \text{definition of } F \\ &= \omega(G(id, \lambda \circ \lambda^{-1}, \lambda)(u_G \otimes_G x)) & \text{isomorphism} \\ &= \omega(G(\lambda^{-1}, \lambda, \lambda)(u_G \otimes_G x)) & \text{dinaturality of } \omega \\ &= \omega x & \text{left identity of } G \end{aligned}$$

The right identity case is similar.

Finally, the associativity law for  $\otimes_F$  follows from the associativity of  $\otimes_G$ :

$$\begin{aligned} F\alpha(\omega x \otimes_F (\omega y \otimes_F \omega z)) & \\ = F\alpha(\omega x \otimes_F \omega(y \otimes_G z)) & \text{definition of } \otimes_F \\ = F\alpha(\omega(x \otimes_G (y \otimes_G z))) & \text{definition of } \otimes_F \\ = \omega(G(id, id, \alpha)(x \otimes_G (y \otimes_G z))) & \text{definition of } F \\ = \omega(G(id, \alpha \circ \alpha^{-1}, \alpha)(x \otimes_G (y \otimes_G z))) & \text{isomorphism} \\ = \omega(G(\alpha^{-1}, \alpha, \alpha)(x \otimes_G (y \otimes_G z))) & \text{dinaturality of } \omega \\ = \omega((x \otimes_G y) \otimes_G z) & \text{associativity of } G \\ = \omega(x \otimes_G y) \otimes_F \omega z & \text{definition of } \otimes_F \\ = (\omega x \otimes_F \omega y) \otimes_F \omega z & \text{definition of } \otimes_F \end{aligned}$$

□

As a further example, we have the coend encoding of the final fixed point  $\nu F$  of a functor  $F : \mathcal{A} \times \mathcal{B} \rightarrow \mathcal{B}$ :

$$\nu Fa \cong \exists b. b \times (b \rightarrow F(a, b))$$



which is a coend of  $G(b', b, a) = b \times (b' \rightarrow F(a, b))$ , and yields the same applicative functor as discussed in Section 3.

## 6 Conclusion

We have established a number of general constructions of lax monoidal functors, and therefore of applicative functors. In examples such as the permutation phrases of Section 5.2, we showed that by combining these constructions we could account for quite complex (and useful) applicative functors, avoiding the need for specific proofs of their laws. By breaking the functors down into simple building blocks, we have clarified their relationships, as well providing the tools to build more applications. The next stage is to examine the possible combinations, and to consider other constructions.

## References

1. Michael Abbott, Thorsten Altenkirch, and Neil Ghani. Categories of containers. In Andrew D. Gordon, editor, *FoSSaCS*, volume 2620 of *LNCS*, pages 23–38, 2003.
2. Robert Atkey. What is a categorical model of arrows? *Electronic Notes on Theoretical Computer Science*, 229(5):19–37, 2011.
3. Arthur I. Baars, Andres Löh, and S. Doaitse Swierstra. Parsing permutation phrases. *Journal of Functional Programming*, 14(6):635–646, 2004.
4. Duncan Coutts. Arrows for errors: Extending the error monad, 2002. Unpublished presentation at the Summer School on Advanced Functional Programming.
5. S. Eilenberg and G.M. Kelly. Closed categories. In S. Eilenberg, D. K. Harrison, H. Röhrl, and S. MacLane, editors, *Proceedings of the Conference on Categorical Algebra*, pages 421–562. Springer, 1966.
6. Conal Elliott. Denotational design with type class morphisms. Technical Report 2009-01, LambdaPix, 2009.
7. Ralf Hinze. Type fusion. In Dusko Pavlovic and Michael Johnson, editors, *Proceedings of the Thirteenth International Conference on Algebraic Methodology And Software Technology (AMAST '10)*, volume 6486 of *LNCS*, pages 92–110. Springer, 2010.
8. Ralf Hinze. Kan extensions for program optimisation, or: Art and dan explain an old trick. In Jeremy Gibbons and Pablo Nogueira, editors, *Mathematics of Program Construction*, 2012.
9. John Hughes. Generalising monads to arrows. *Science of Computer Programming*, 37(1–3):67–111, May 2000.
10. Patricia Johann and Neil Ghani. A principled approach to programming with nested types in Haskell. *Higher-Order and Symbolic Computation*, 22(2):155–189, 2009.
11. G. M. Kelly. *Basic concepts of enriched category theory*, volume 64 of *London Mathematical Society Lecture Note Series*. Cambridge University Press, 1982.
12. J. Lambek and P. J. Scott. *Introduction to Higher Order Categorical Logic*. Number 7 in Cambridge Studies in Advanced Mathematics. Cambridge University Press, Cambridge, 1986.
13. Saunders Mac Lane. *Categories for the Working Mathematician*. Springer, New York, 1971.

14. Kevin Matlage and Andy Gill. Every animation should have a beginning, a middle, and an end: A case study of using a functor-based animation language. In Rex Page, Zoltán Horváth, and Viktória Zsók, editors, *Proceedings of the 11th International Conference on Trends in Functional Programming*, volume 6546 of *LNCS*, pages 150–165. Springer, 2011.
15. Conor McBride and Ross Paterson. Applicative programming with effects. *Journal of Functional Programming*, 18(1):1–13, 2008.
16. Eugenio Moggi. Computational lambda-calculus and monads. In *Logic in Computer Science*, pages 14–23. IEEE Computer Society Press, 1989.
17. Eugenio Moggi. An abstract view of programming languages. Technical Report ECS-LFCS-90-113, Laboratory for Foundations of Computer Science, University of Edinburgh, 1990.
18. S. Doaitse Swierstra and Luc Duponcheel. Deterministic, error-correcting combinator parsers. In John Launchbury, Erik Meijer, and Tim Sheard, editors, *Advanced Functional Programming*, volume 1129 of *LNCS*, pages 184–207. Springer, 1996.
19. Janis Voigtländer. Asymptotic improvement of functions over free monads. In Philippe Audebaud and Christine Paulin-Mohring, editors, *Mathematics of Program Construction*, volume 5133 of *LNCS*, pages 388–403. Springer, 2008.
20. Philip Wadler. Comprehending monads. *Mathematical Structures in Computer Science*, 2(4):461–493, 1992.